


# Entfernungsberechnung und Antennenrichtung mit QTH-Locator

## Berechnungsmethode für Excel

Waldemar Breiling, DJ1DZ

**E**s gibt viele Programme, die es ermöglichen, aus zwei QTH-Kennern die Entfernung zu berechnen. Wollte ich diese Entfernung in mein QSO-Logbuch eintragen, so schreibe ich die zwei QTH-Locator in das Programm und das Ergebnis in mein Logbuch. Mittlerweile ist mein Logbuch auf mehr als 1000 Einträge gewachsen. Hier zeigt sich schon, dass obige Methode nicht infrage kommt. Ziel ist es nun, natürlich über Excel, eine Berechnungsmethode zu finden, die alles automatisch ablaufen lässt. Dazu ist es notwendig, den QTH-Kenner in Längen und Breitengrade umzurechnen und die Erde als Kugel mit einem Radius von 6371 km zu betrachten. Der Radius ist ein anerkannter Mittelwert aus Äquatorradius und Polarradius.



**Aufmacherbild (oben):** Hier zeigt der Autor, wie Entfernung und Antennenrichtung, nur aus dem QTH-Locator der Gegenstation, berechnet werden kann.

*Design by Wolfgang.Lüdtke, Kiel*


## QTH-Locator System

Der QTH-Kenner beschreibt Größtfelder, Großfelder, und Kleinfelder.

Beispiel: JO31PL – JO = Größtfeld, 31 = Großfeld, PL=Kleinfeld.

Die Größtfelder (**Bild 1**) sind wie folgt aufgeteilt: Der erste Buchstabe ist A und zählt am Südpol nach Osten bis R. Also der erste Buchstabe ist für die Einteilung der Erde nach Osten, somit A-R (360°) in 18 Reihen.

**Bild 1 (rechts):**  
Die Größtfelder in 18 x 18 = 324 Teile sind über den gesamten Globus verteilt, je 10° Nord und 20° Ost




Der 2. Buchstabe fängt auch mit A an und zieht vom Südpol bis zum Nordpol, ebenfalls bis R, also (180°) auch 18 Reihen.

Teilt man jetzt 360 Grad in 18 Teile, so ergibt sich für jedes Großfeld 20 Längengrade. Die 180 Breitengrade werden auch in 18 Teile zu je 10 Grad unterteilt. Legt man diese Einteilung so, dass Äquator und 0-Meridian (durch Greenwich, England) diese teilen, stellt sich nachfolgendes **Bild 1** ein.

Das Größtfeld JO aus unserem Beispiel ist markiert.

Die Größtfelder in 18 x 18 = 324 Teile sind über den gesamten Globus verteilt, je 10° Nord und 20° Ost. Jedes Größtfeld wird wiederum aufgeteilt in 10 x 10 = 100 Großfelder (**Bild 3**).

**Bild 2 (rechts):**  
Das Größtfeld JO ist in 10 x 10 = 100 Großfelder aufgeteilt, je 1° Nord und 2° Ost



	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q	R	S	T	U	V	W	X
X	AX	BX	CX	DX																				
W	AW	BW	CW																					
V																								
U																								
T																								
S																								
R																								
Q																								
P																								
O																								
N																								
M																								
L																								
K																								
J																								
I																								
H																								
G																								
F																								
E																								
D																								
C																								
B																								
A	AB	BB	CB																					
	AA	BA	CA	DA																				

**Bild 3 rechts:**

Das Großfeld ist in  $24 \times 24 = 576$  Kleinfelder aufgeteilt, je  $1^\circ / 24 = 0,04166^\circ$  Nord und  $2^\circ / 24 = 0,0833^\circ$  Ost

Der untere Rand von JO liegt, gemessen vom Äquator bei  $50^\circ$ , und der linke Rand, gemessen vom 0-Meridian, bei  $0^\circ$ . Für unser Beispiel ist das Feld 31 markiert. Da jedes Großfeld  $2^\circ$  entspricht, zählen wir vom linken Rand (0-Meridian) bis zum Feld (31)  $6^\circ$  nach Osten. Vom unteren Rand (JO liegt bei  $50^\circ$ ) sind es noch  $1^\circ$  in nördliche Richtung, also insgesamt  $50^\circ + 1^\circ = 51^\circ$  Nord.

Das Großfeld ist in  $24 \times 24 = 576$  Teile aufgeteilt. Je  $1^\circ/24 = 0,04166^\circ$  Nord und  $2^\circ/24 = 0,0833^\circ$  Ost.

Betrachten wir die Mitte vom Kleinfeld PL (Bild 3), dann zählen wir vom linken Rand A bis P-Mitte, 15,5 Teile. Das sind  $15,5 \times 0,0833^\circ = 1,292^\circ$  Ost. Vom 0-Meridian gemessen  $6^\circ + 1,292^\circ = 7,292^\circ$  Ost.

In nördliche Richtung sind es von A bis Mitte L 11,5 Teile also  $11,5 \times 0,04166^\circ = 0,4791^\circ$  Nord. Gemessen vom Äquator liegen noch 5 Großfelder mit je  $10^\circ$  Nord und 1 Großfeld mit  $1^\circ$  Nord dazwischen. Also  $5 \times 10^\circ + 1 \times 1^\circ + 0,4791^\circ = 51,4791^\circ$

Der QTH-Locator JO31PL liegt somit bei  $7,292^\circ$  Ost und  $51,4791^\circ$  Nord.

## Und nun zu Excel

Um in Excel rechnen zu können, benötigen wir Zahlen. Da alle Buchstaben, Zahlen und Sonderzeichen in einem ASCII-Code stecken, benutzen wir diesen.

Der Buchstabe A hat den Wert 65 (ausprobieren mit Alt 65). J hat den Wert 74 (Alt 74).  $74 - 65 = 9$ . Die Zahlen beginnen mit 0 bei 48 bis 9 bei ASCII-Code 57.

Mit dieser Erkenntnis bestimmen wir die Buchstaben bzw. Zahlen J;O;3;1;P;L.

Der Berechnungswert für J ist 9

Der Berechnungswert für O ist 14

Der Berechnungswert für 3 ist 3

Der Berechnungswert für 1 ist 1

Der Berechnungswert für P ist 15

Der Berechnungswert für L ist 11

Steht der QTH-Locator in B2, so sieht die Formel in Excel für die Buchstaben und Zahlen wie folgt aus:

```
=(CODE(TEIL(B2;1;1))-65) 1.Buchstabe J
=(CODE(TEIL(B2;2;1))-65) 2.Buchstabe O
=(CODE(TEIL(B2;3;1))-48) 3.(Zahl) 3
=(CODE(TEIL(B2;4;1))-48) 4.(Zahl) 1
=(CODE(TEIL(B2;5;1))-65) 5.Buchstabe P
=(CODE(TEIL(B2;6;1))-65) 6.Buchstabe L
```

**Bild 4** zeigt die Verteilung für die Berechnung der Längen- und Breitengrade und gibt Hinweis auf den Multiplikator (für die Längengrade  $20^\circ$ ,  $2^\circ$ ,  $(2 / 24)^\circ$  und für die Breitengrade  $10^\circ$ ,  $1^\circ$ ,  $(1 / 24)^\circ$ ).

Für den Längengrad nehmen wir die **1.** Stelle (9) und multiplizieren mit  $20^\circ$ . Mit  $-180^\circ$  stellen wir fest, liegen wir östlich oder westlich vom 0-Meridian. Die **3.** Stelle (3) multiplizieren wir mit  $2^\circ$  und die **5.** Stelle (15) multiplizieren wir mit  $1^\circ / 12$ . Jetzt addieren wir alle drei Ergebnisse.

$$9 \times 20 - 180 + 3 \times 2 + 15,5 \times 2 / 24 = 7,29^\circ$$

```
=((CODE(TEIL(B2;1;1))-65)*360/18-180)+
((CODE(TEIL(B2;3;1))-48)*2)+
((CODE(TEIL(B2;5;1))-65+0,5) *2/24)
```


Für den Breitengrad die **2., 4. und 6.** Stelle, also **O, 1, L**. Die 2. Stelle wird mit  $10^\circ$  multipliziert. Mit  $-90$  wird nördlich oder südlich festgestellt.

$$14 \times 10 - 90 + 1 \times 1 + 11,5 \times 1 / 24 = 51,48^\circ$$

```
=((CODE(TEIL(B2;2;1))-65)*180/18-90)+
((CODE(TEIL(B2;4;1))-48)*1)+
((CODE (TEIL(B2;6;1))-65+0,5)*1/24)
```

Ergebnis ist  $7,292^\circ$  Ost;  $51,4791^\circ$  Nord. Negative Werte sind für West und Süd.

Jetzt haben wir die nötigen Winkel. Zusammen mit dem Erdradius (6371km) können wir nun die Entfernungen zu den Zielkoordinaten bestimmen.



**Bild 4:** Die Verteilung der Längen- und Breitengrade zeigen die Multiplikationen der Buchstaben und Zahlen.

## Entfernungsberechnung

Berechnet werden soll die Entfernung von unserem QTH-Locator (QTH1) zum Ziel-Locator (QTH2).

### Berechnungsbeispiel:

QTH1 = JO31PL, zu QTH2 = HP23FG

Koordinaten wie vor ermittelt ergibt:

QTH1 (JO31PL): 7,29° Ost; 51,4791° Nord

QTH2 (HP23FG): -35,54° West; 63,27° Nord

(die Werte für West und Süd sind negativ)

Zuerst betrachten wir QTH1. Schneiden wir längs des Längengrades (7,29°) die Erdkugel auf, entsteht das

#### 1. Dreieck (Bild 5)

Winkel  $a_1 = 51,48^\circ$

Radius  $r = 6\,371\,000\text{ m}$

$h_1 = s \cdot \sin(a_1) \times r = 4\,984\,554,14\text{ m}$

$h_1 = 4984,55\text{ km}$

$L_1 = h_1 / \tan(51,48^\circ) = 3\,967\,853,45\text{ m}$

$L_1 = 3967,85\text{ km}$




Bild 5: 1. Dreieck – Schnitt durch den Längengrad 7,29° bei QTH1

Ebenso bei QTH2 den Längengrad bei -35,54° schneiden.

So entsteht das

#### 2. Dreieck (Bild 6)

Winkel  $a_2 = 63,27^\circ$

Radius  $r = 6\,371\,000\text{ m}$

$h_2 = \sin(a_2) \times r = 5\,690\,211,15\text{ m}$

$h_2 = 5690,21\text{ km}$

$L_2 = h_2 / \tan(a_2) = 2\,865\,508,34\text{ m}$

$L_2 = 2865,51\text{ km}$




Bild 6: 2. Dreieck – Schnitt durch den Längengrad -35,54° bei QTH2

Nun schneiden wir die Erdkugel durch den Äquator und projizieren QTH1 und QTH2 auf die Schnittfläche.

### 3. Dreieck (Bild 7)

$$L1 = 3967,85 \text{ km}$$

$$L2 = 2865,51 \text{ km}$$

$$\text{Winkel } a3 = \text{abs}(7,29^\circ - (-35,54^\circ))$$

$$a3 = 42,83^\circ$$

$$L3^2 = L1^2 + L2^2 - (2 \times L1 \times L2 \times \cos(a3))$$

Nach  $\sqrt{\quad}$  (Wurzel ziehen)

$$L3 = 2\,697\,979,80 \text{ m}$$

$$L3 = 2\,697,98 \text{ km}$$




Bild 7: 3. Dreieck – Schnitt durch Äquator. QTH1 und QTH2 auf Äquatorebene projiziert

Um das 4. Dreieck zu erzeugen, schneiden wir durch QTH1 und QTH2, parallel zur Erdachse

### 4. Dreieck (Bild 8)

$$L3, L4, L5$$

$$h1 = 4\,984\,554,14 \text{ m}$$

$$h2 = 5\,690\,211,15 \text{ m}$$

$$L3 = 2\,697\,979,80 \text{ m}$$

$$L4 = \text{abs}(h1 - h2)$$

$$L4 = 705\,657,01 \text{ m}$$

$$L5^2 = L3^2 + L4^2$$

$$L5 = 2\,788\,735,70 \text{ m}$$

L5 ist die direkte Verbindung QTH1, QTH2 ohne Berücksichtigung der Erdkrümmung.




Bild 8: 4. Dreieck – Schnitt durch QTH1 und QTH2, parallel zur Erdachse

Unser nächster und letzter Schnitt durchläuft die Punkte QTH1, QTH2 und M (Erdmittelpunkt).

### 5. Dreieck (Bild 9)

$$\text{Radius } r = 6\,371\,000 \text{ m}$$

$$L5 = 2\,788\,735,70 \text{ m}$$

$$\sin(a5/2) = (L5/2)/r$$

$$a5 = \arcsin((L5/2)/r) \times 2$$


$$B0 = (2r \times p) / 360 \times a5$$

$$B0 = (2r \times p) / 360 \times (\arcsin((L5/2)/r) \times 2)$$

$$B0 = 2\,811\,493,35 \text{ m}$$

Bild 9 (rechts): 5. Dreieck

Schnitt durchläuft QTH1, QTH2 und Erdmittelpunkt



**B0 = 2811,49 km ist unser Ergebnis, errechnet von QTH1 bis QTH2 über die Erdkrümmung.**

## Antennenrichtung

**N**ach unserem Beispiel: QTH2 HP23FG.

Bei der Betrachtung unserer Ergebnisse entsteht für die Richtung QTH1 nach QTH2 ein sphärisches Dreieck (**Bild 10**).

Ermittelt werden soll der Winkel B-A-C (Nordpol-QTH1-QTH2). Angewendet wird der Seitenkosinussatz des nachfolgenden sphärischen Dreiecks.

$$\cos(a) = \cos(b) \times \cos(c) + \sin(b) \times \sin(c) \times \cos(\alpha) \quad [\alpha - \text{Alpha}]$$

$$\cos(\alpha) = (\cos(a) - \cos(b) \times \cos(c)) / (\sin(b) \times \sin(c))$$

Der Winkel (a) ist der Ergänzungswinkel QTH2 zum Nordpol  $90^\circ - 63,27^\circ = 26,73^\circ = a$

Der Winkel (c) ist der Ergänzungswinkel QTH1 zum Nordpol  $90^\circ - 51,48^\circ = 38,52^\circ = c$

Der Winkel (b) ergibt sich aus dem Radius und gradliniger Entfernung QTH1 nach QTH2 (ist mit Winkel (a5) berechnet), siehe 5. Dreieck (Bild 9).

$$a5 = \arcsin((L5/2) / r) \times 2$$

$$B0 = (2 \times \text{Radius} \times p) / 360 \times a5$$

$$a5 = b$$

$$b = B0 \times 360 / (2 \times \text{radius} \times p)$$


$$b = 25,29^\circ$$

Die Winkel in obige Formel eingesetzt:

$$\cos(\alpha) = (\cos(26,73^\circ) - \cos(25,29^\circ) \times \cos(38,52^\circ)) / (\sin(25,29^\circ) \times \sin(38,52^\circ))$$

$$\cos(\alpha) = 0,6981$$

$$\alpha = \arccos(\alpha) \quad \alpha = 45,73^\circ$$



**Bild 10: Die Punkte auf dem Globus QTH1, QTH2 und der Nordpol bilden das sphärische Dreieck**

Die Rotorsteuergeräte drehen in der Regel von Nord  $0^\circ$  rechts nach Ost, Süd, West bis  $360^\circ$  Nord.

Jetzt gilt noch die Bedingung: Liegt die Station QTH2 links (westlich) von QTH1, dann muss von  $360^\circ$  der errechnete Wert abgezogen werden. Das ist immer dann der Fall, wenn der Längengrad QTH2 kleiner ist als der von QTH1.

**Die einzustellende Richtung ist also  $360^\circ - 45,73^\circ = 314,27^\circ$ .**

## Und nun zu den Excel-Formeln

Über die QTH-Kenner sollen die Längen und Breitengrade ermittelt werden. Diese Winkelangaben werden nun für die Erstellung einer einzigen Formel genutzt. In Excel würde das dann etwa so wie in **Bild 11** aussehen.

Alle Gradangaben in der Spalte C, die negativ sind, befinden sich westlich vom Nullmeridian.

Alle Gradangaben in der Spalte D, die negativ sind, befinden sich südlich des Äquators.

Das Feld C2 wird mit **HomeE** bezeichnet

Das Feld D2 wird mit **HomeN** bezeichnet

Das Feld E2 wird mit **Radius** bezeichnet  
(Formeln, Namensmanager)

#### Formel für C2:

eigener QTH-Kenner (E/W)

$$=((\text{CODE}(\text{TEIL}(\text{B2};1;1))-65)*360/18-180)+((\text{CODE}(\text{TEIL}(\text{B2};3;1))-48)*2)+((\text{CODE}(\text{TEIL}(\text{B2};5;1))-65+0,5)*2/24)$$

Analog dazu für den Ziel-QTH-Kenner, die Felder C5, C6, C7 usw.

Negativer Wert = West

Positiver Wert = Ost

#### Formel für D2:

eigener QTH-Kenner (N/S)

$$=((\text{CODE}(\text{TEIL}(\text{B2};2;1))-65)*180/18-90)+((\text{CODE}(\text{TEIL}(\text{B2};4;1))-48)*1)+((\text{CODE}(\text{TEIL}(\text{B2};6;1))-65+0,5)*1/24)$$

Analog dazu für den Ziel-QTH-Kenner, die Felder D5, D6, D7 usw.

Positiver Wert = Nord

Negativer Wert = Süd

#### Formel für E5: Entfernung

$$\begin{aligned} &= \text{Radius}/1000*2*\text{PI}()/360*\text{ARCSIN}(\text{WURZEL}(\text{WURZEL}((\text{SIN}(\text{BOGENMASS}(\text{HomeN}))*\text{Radius}/\text{TAN}(\text{BOGENMASS}(\text{HomeN})))^2+(\text{SIN}(\text{BOGENMASS}(\text{D5}))*\text{Radius}/\text{TAN}(\text{BOGENMASS}(\text{D5})))^2- \\ &2*(\text{SIN}(\text{BOGENMASS}(\text{HomeN}))*\text{Radius}/\text{TAN}(\text{BOGENMASS}(\text{HomeN}))) * \text{SIN}(\text{BOGENMASS}(\text{D5}))*\text{Radius}/\text{TAN}(\text{BOGENMASS}(\text{D5}))) \\ &*\text{COS}(\text{BOGENMASS}(\text{ABS}(\text{HomeE}-\text{C5}))))^2+((\text{SIN}(\text{BOGENMASS}(\text{D5}))*\text{Radius})-\text{SIN}(\text{BOGENMASS}(\text{HomeN}))*\text{Radius})^2)/2/\text{Radius}) * 180/\text{PI}()*2 \end{aligned}$$

Analog dazu die Felder E6, E7, E8 usw.

#### Formel für G5: Richtung

$$\begin{aligned} &= \text{ABS}(360*((\text{C5}-\text{HomeE})-\text{ABS}(\text{C5}-\text{HomeE}))/(\text{C5}-\text{HomeE})/2)-\text{ARCCOS}((\text{COS}(\text{BOGENMASS}(90-\text{D5}))- \\ &\text{COS}(\text{BOGENMASS}(\text{E5} * 360/2/\text{Radius}/\text{PI}()*1000)) * \text{COS}(\text{BOGENMASS}(90-\text{HomeN}))) / \text{SIN}(\text{BOGENMASS}(\text{E5} * 360/2/\text{Radius}/\text{PI}()*1000)) / \text{SIN}(\text{BOGENMASS}(90-\text{HomeN}))) * 180/\text{PI}()) \end{aligned}$$

Analog dazu die Felder G6, G7, G8 usw.

	A	B	C	D	E	F	G
1	Call	QTH	E/-W	N/-S	Erdradius		
2	DJ1DZ	JO31PL	7,292	51,479	6371000,785		
3							
4	Cal	QTH	E/	N	Entfernu	Richtu	
5		HP23FG	-35,5417°	63,2708°	2.811 km	314,3°	
6		PM95DK	138,2917°	35,4375°	9.236 km	38,3 °	
7		KP10ML	23,0417°	60,4792°	1.394 km	38,0 °	
8		IN83LP	-3,0417°	43,6458°	1.164 km	225,6 °	
9		KP01ML	21,0417°	61,4792°	1.390 km	31,6 °	
10		PG22LM	124,9583°	-27,4792°	14.248 km	87,5 °	
11		PM95AA	138,0417°	35,0208°	9.266 km	38,7 °	
12		KP24ND	25,1250°	64,1458°	1.748 km	29,5 °	
13		JO02LP	0,9583°	52,6458°	452 km	289,2 °	
14		IN83LM	-3,0417°	43,5208°	1.175 km	225,2 °	
15		GH92BM	-41,8750°	-17,4792°	9.026 km	226,9 °	
16		IM99LN	-1,0417°	39,5625°	1.473 km	209,2 °	
17		OO22RM	105,4583°	52,5208°	6.166 km	47,0 °	
18		LL93KE	58,8750°	23,1875°	5.383 km	105,6 °	
19		HP23FG	-35,5417°	63,2708°	2.811 km	314,3 °	
20		ON67AA	112,0417°	47,0208°	6.931 km	48,1 °	
21		GH22GH	-55,4583°	-17,6875°	9.791 km	237,9 °	
22		KP10ML	23,0417°	60,4792°	1.394 km	38,0 °	
23		JM37SQ	7,5417°	37,6875°	1.534 km	179,2 °	
24		JO43IU	8,7083°	53,8542°	281 km	19,3 °	
25		JO31QS	7,3750°	51,7708°	33 km	10,0 °	
26		JP31QS	7,3750°	61,7708°	1.144 km	,2 °	
27		JN90DA	18,2917°	40,0208°	1.530 km	142,1 °	

**Bild 11: Ein Excel-Beispiel zur Anordnung der Berechnungsfelder**

Links:

- [1] <https://de.wikipedia.org/wiki/QTH-Locator>
- [2] Martin Leuschner <http://dk3ml.de/2016/05/03/qth-locator-karte/>
- [3] <https://de.wikipedia.org/wiki/Erdradius>
- [4] Norbert Friedrichs <http://dl6mmm.darc.de/QTH-Locator.pdf>